
à la Mondrian

Abstract
Mondriaan has worked most of his live as an abstract painter, influenced by the magic real-
ism of Jan Toorop, and by Cubism and Pointillism. He was member of De Stijl and has lived
in Paris and in New York. Some of his work seems to have been composed randomly, though
he was very precise, as witnessed by the overpainting of various squares in his Victory Boo-
gie-Woogie. Mondriaan's `random' work Composition in Line 1916, is emulated and varied
in MetaPost and PostScript, in color, with the lines (position and size) randomly chosen. He
was the first painter to frame work by Lozenges. Division of the sides of his Lozenge with
2 lines is near to the golden ratio. Emulated Lozenges obeying the golden ratio have been
included. The variations look nevertheless Mondriaan-esque.

Art, color, cubism, De Stijl, golden ratio, Mondrian, pointillism, pseudo-random numbers,

Introduction
My applied math professor Hans Lauwerier1 published in 1987, long after I fin-
ished my math education, Fractals—meetkundige figuren in eindeloze herhaling. It
contained the following picture, which he called ‘à la Mondriaan’2

Lauwerier used this picture to illustrate a very simple algorithm3 for generating
a pseudo-random number sequence in BASIC

“... start with a four digit number, the seed, square it and delete the first and
last two numbers, and repeat the process of squaring and deleting ...”

He randomly positioned (place, horizontally or vertically) the line pieces of random
length. I'll improve on his picture with the use of a shade of color, and variation in
line thickness. Moreover, I'll cadre by different frames.

Mondriaan
Mondriaan was born in 1872 at Amersfoort and started drawing à la nature.

Meisje 1890 Winterswijk 1895

Jan Toorop has influenced Mondriaan by his new realism. Mondriaan was often a
guest at Toorop's in Zeeland in the beginning of the 20th century; the (light at the)
seashore inspired Mondriaan.

Zeeland farmer 1909 Red Cloud 1907

Arum Lilies 1909 Lighthouse WC 1908/9

Mondriaan exercised an oval as a boundary, not as a frame. The Composition in
Oval 1913 reminds me of the work of George Braque.

Composition in Oval 1913 Composition in Oval 1914

He has lived in Paris, 1919-1938, and was influenced by cubism and pointillism.

Blossoming Apple Tree 1912 Beach with one Pier 1909

I was familiar with some of Mondriaan's works from the ‘De Stijl’4 period, where
he exercised the use of primary color panes and straight lines, which reminds me of
stained glass windows.

emulation: Large Composition with Red, Blue and Yellow 1928

His series of Lozenge's from the twenties, ended in a minimal one, given below at
the right.

Lozenge with 3 lines 1925 Lozenge with 2 lines 1931

I measured his Lozenges and was suprised that he missed the divison of the sides
by the golden ratio: his sides are divided in 3.9 : 5.6; the golden ratioΦ does not hold,
ie 3.9 : 5.6 ≠ 5.6 : 9.5. A tiny difference, but nonetheless.

Φ emulation Φ emulation

%Mondriaan-like Lozenge with 3 lines. Fall 2009 kisa1@xs4all.nl
beginfig(0)
s=200;
z1=(0,.5s); z2=(.5s,s); z3=(s,.5s); z4=(.5s,0);
z5=0.618/1.618[z2,z3]; z6=0.618/1.618[z3,z4];
z7= 1/1.618[z3,z4]; z8=1/1.618[z4,z1];
z9=0.382/1.618[z1,z4]; z10=0.382/1.618[z1,z2];
z11= (z5--z7) intersectionpoint (z6--z8);
path p; p = z1--z2--z3--z4--cycle;
pw=4; pickup pencircle scaled pw;
fill p withcolor .95white;
fill z1--z10--z9--cycle withcolor blue;
fill z7--z11--z6--cycle withcolor red+green;
fill z8--z11--z7--z4--cycle withcolor .85white;
draw z5--z7;
draw z6--z8;
draw z9--z10;
clip currentpicture to p;
draw p withcolor .97white;
endfig
end

For the right Lozenge the main part of the above reads

fill p withcolor .95white;
draw z5--z7;
draw z6--z8;
clip currentpicture to p;
draw p withcolor .97white;

Remark. In contrast with PostScript we don't have to translate the origin inMetaPost
to somewhere in the middle.
For my last year's birthday Sveta and I composed the following à la Mondriaan invi-
tation starting from the left Φ-Lozenge.

Not so long ago, I used one of his Apple trees, along with a little poetic proza, for
our local gardeners bulletin. It is about one of our Apple trees, which in every season
has something to offer:

spring, blossoming beauty
summer, healthy shade
autumn, fruits to enjoy
winter, beautiful silhouet à la Mondriaan.

His unfinished Victory Boogie-Woogie is shown on the cover of the biography about
Mondriaan.

Mondriaan has the reputation to be very precise, and I was curious whether some of
his works

“… could be generated randomly in color and nevertheless convey aMondriaan
impression … ”

although at the time he did not use color for the ‘random’ pictures given on the next
page.

Composition in Line, 1916 Composition 10, 1915

Variation of Mondriaan's ‘random' art
Is it random?

As far as I understand it, his work was composed precisely, very precisely, as
witnessed by his unfinished Victory Boogie-Woogie, where several squares have
been painted over.

Square, Oval and Lozenge cadres. I generated an abstract picture of lines, with a
spread in size, thickness and with a color shade, such that the impression is blue-like,
green-like,...

I intended these pictures as presents to my friends. Made them unique by using
the date of birth (month and day) as seed for the pseudo-random number generator
in MetaPost.5 Colored the frame with their favorite color, which also biases the color
shade.

Below is Sveta's one in pink, with 1007 as seed for the pseudo-random number
generator.

cgl okt09

How to?
First, we have to decide on the size of the rectangle. Next the number of randoms.

I chose 500, but that can be adjusted at will. The lines are drawn within a rectangle
of 180 by 180 ⋅ 1.618, obeying the golden ratio. For the position in the rectangle the

random number generator must generate numbers between 0 0 180 ≈291, conform
PostScript's BoundingBox convention to denote a rectangle.

uniformdeviate delivers a number between 0 and its argument. We have the in-
tervals (0, u) and (0,v), so the invokes for the (x, y)-position read

hx:= uniformdeviate u; hy:= uniformdeviate v;

I alternate between vertical and horizontal lines, where the lines have maximum
length size=10, and maximum (pen)width pw=5pt. The actual size, horizontally as
well as vertically, is determined by an invoke of uniformdeviate

draw (hx, hy)--(hx + uniformdeviate size, hy) withcolor rgb%vertical line

Line thickness is varied as follows:

pickup pencircle scaled ((uniformdeviate 1)*pw);

The most difficult part is the color, which I let vary a little around the chosen value,
in order to generate a shade but … gives the main color impression. I chose a multi-
plicative factor uniformdeviate 2 and called this spread.

To finish it up, I clip and border first to the rectangle, next to the oval, and finally
to the diamond; all border increasingly the same pattern.

%Mondriaan-alike. Fall 2009 CGL
if scantokens(mpversion) > 1.005: outputtemplate :=
else: filenametemplate
fi "%j.eps";
prologues:=3;
beginfig(0)
u:=180; %rectangle u x v, golden ratio proportion
v:=1.618u;
size=10; pw=5;
n=500; %number of randoms
path p.r, %rectangle

p.o, %oval
p.d; %diamond

picture pic.r, pic.o, pic.d, signature;
defaultfont:="ptmr8r";
label.rt("cgl okt09", (2.75u+4pw, 2pw));%seed may be shown instead
signature:=currentpicture; currentpicture:=nullpicture;
%
color rgb, colorofchoice;
%parameters for Sveta
randomseed:=1007; colorofchoice:= .8*red +.2*green .7*blue;%roze
%
for j=0 upto n:
pickup pencircle scaled ((uniformdeviate 1)*pw*pt); linecap:=squared;
hx:= uniformdeviate u; hy:= uniformdeviate v;
rgb:= .8*(uniformdeviate 2)*red + .2*(uniformdeviate 2)*green

+ .7*(uniformdeviate 2)*blue;
draw (hx,hy)--(hx,hy+ uniformdeviate size) withcolor rgb;
hx:= uniformdeviate u; hy:= uniformdeviate v;
rgb:= .8*(uniformdeviate 2)*red + .2*(uniformdeviate 2)*green

+ .7*(uniformdeviate 2)*blue;

draw (hx,hy)--(hx+uniformdeviate size, hy) withcolor rgb;
endfor;
p.r= ((0, v)--(u,v)--(u,0)--(0,0)--cycle) shifted (2pw,2pw);
p.o= ((.5u,v){right}...(u,.5v){down}...(.5u,0){left}...(0,.5v){up}

...cycle) shifted (u+4pw,2pw);
p.d= ((.5u,v)--(u,.5v)--(.5u,0)--(0,.5v)--cycle) shifted (2u+6pw,2pw);
currentpicture:=currentpicture shifted (2pw,pw);
clip currentpicture to p.r;
pic.o:=currentpicture shifted (u+4pw,0);
clip pic.o to p.o;%clip more
pic.d:= currentpicture shifted (2u+6pw,0);
clip pic.d to p.d;%clip even more
addto currentpicture also pic.o;%add shifted pictures
addto currentpicture also pic.d;
%
pickup pencircle scaled pw; linecap:=rounded;
drawoptions (withcolor .8*red +.2*green .7*blue);
draw p.r; draw p.o; draw p.d;
addto currentpicture also signature;
endfig end

I reused this program for generating personalized presents several times, but I am
happier with the PostScript operator given below. When the picture was for a man,
I copied the Square or the Lozenge, and when it is for a woman I copied the Oval.6

Frame the picture in a physical frame and ready it is, apart from a wrap-around
paper.

Note that all three frames have the same pattern: the pattern is clipped by frame
variations. Subtle is the printing of the signature outside the clipping area.

PostScript operator variant
Why a PostScript variant of the MetaPost program?

allows larger numbers (seed eg 22121943)
more convenient clipping
operator (library)
no invoke of MP
easier experimenting with color shade
improved code.

In this variant I generate only one framed picture triggered by the value 0, 1, or 2 for
the last parameter on the stack, where 0=Square 1=Oval 2=Lozenge. The size is 420
by 420 ⋅ 1.618 and positioned with the lower left corner at

100 50 translate

for a centered result. The complete birthday date, ddmmyyyy, can be supplied. The
code is not a direct translation, because of the language differences and because I
matured in coding and was after a library operator. The thickness of the strokes are
obtained as follows

/w {maxwidth unifrmdev} def
... w setlinewidth

The line pieces are positioned symmetrically

/laux l 2 div def
xaux laux sub yaux moveto xaux laux add yaux lineto w setlinewidth
color setrgbcolor stroke%h-line

The spread in color has been implemented as: if an rgb-value = 0, it is changed into
eps=0.1 and the resulting value c.q. the original non-zero value is multiplied by the
spread factor.

/spread {2 unifrmdev mul }def%interval (0, 2)
/color{r 0 eq {eps}{r} ifelse spread

g 0 eq {eps}{g} ifelse spread
b 0 eq {eps}{b} ifelse spread} def

The strokes are in a shade of the desired color. The rgb-values are printed at the
top, and the birthday date, the seed for the pseudo-random number generator, at the
bottom.

The most Mondrian-like is maybe the Easter Egg.

/Mondrian
%birthday: ddmmyyyy, a number as seed for srand
%three numbers from the closed interval [0, 1], for the rgb-color values:
% red green blue
%number for the kind of frame (0=Square 1=Oval 2=Lozenge}: 0, 1 or 2;
%==>
%generated Mondrian-alike
{0 begin gsave %savety for not changing the graphics state outside
/form exch def
/b exch def /g exch def /r exch def /date exch def
date srand% start random generator with (birthday date) seed
100 50 translate
%wired-in parameters
/u 420 def /v u 1.618 mul def /hu u 2 div def
/hv v 2 div def% 0 0 u v BB of rectangle

/maxrandom 500 def /maxlength 20 def /maxwidth 3 def /eps 0.1 def
/hx {u unifrmdev} def
/hy {v unifrmdev} def
/l {maxlength unifrmdev}def
/w {maxwidth unifrmdev} def
/spread {2 unifrmdev mul }def %(0, 2)
%/spread {2 unifrmdev 1 add 2 div mul }def%(0.5, 1.5)
%/spread {2 unifrmdev 3 add 4 div mul }def%(0.75, 1.25)
/color{r 0 eq {eps}{r} ifelse spread

g 0 eq {eps}{g} ifelse spread
b 0 eq {eps}{b} ifelse spread} def

form 0 eq {/contour
{0 0 moveto u 0 lineto u v lineto 0 v lineto closepath} def} if %Square

form 1 eq {/contour
{hu hv hu hv 0 360 ellipse} def} if %Oval

form 2 eq {/contour
{hu 0 moveto u hv lineto hu v lineto 0 hv lineto closepath}
def} if %Lozenge

%
gsave contour clip%random pattern will only show up in (is clipped to)

contour maxrandom{%draw pattern in loop confined to contour
/xaux hx def /yaux hy def%position in u x v rectangle
/laux l 2 div def
xaux laux sub yaux moveto xaux laux add yaux lineto w setlinewidth
color setrgbcolor stroke%h-line

/xaux hx def /yaux hy def/laux l 2 div def
xaux yaux laux sub moveto xaux yaux laux add lineto w setlinewidth
color setrgbcolor stroke%v-line

}repeat
grestore %end clipping path
contour 7 setlinewidth r g b setrgbcolor stroke%original color od choice
H12pt setfont /nstr 8 string def
u 85 sub v 10 add moveto (RGB: ) show

r nstr cvs show ( ) show
g nstr cvs show ( ) show
b nstr cvs show

u 85 sub 0 moveto (Seed: ) show date nstr cvs show
grestore end}def%end Mondrian
/Mondrian load 0 26 dict put

A variant suited for cmyk-color values, also some 50 lines, took me about 10 min-
utesje . When k = 0 I did not let it contribute to the spread. A brainfag is that cyan
absorbs red, etc, an approach different from the rgb-model.

Example of use.

%!!PS Mondriaan-achtig. CGL April 2010
%BoundingBox: 0 0 620 790
(C:\\PSlib\\PSlib.eps) run
22121943 1 0 0 0 Mondrian showpage
22121943 0 1 0 1 Mondrian showpage
22121943 0 0 1 2 Mondrian showpage
22121943 .5 .5 .5 2 Mondrian showpage
%%EOF

No invoke of the MetaPost preprocessor is needed, just the use of distiller (as part of
Adobe's Acrobat) or the ps2pdf batch program, or … will yield the visual result.

Conclusion
Mondriaan was undoubtedly precise, but some of his ‘random’-like works can be
created by the use of a pseudo-random number generator and yield nevertheless a
Mondriaan impression.

The positioning of the lines in his Lozenge with 2 lines and Lozenge with 3
lines is also precise, but … division of the sides by the golden ratio Φ yields
equally-well artistic results.

I have input his Victory Boogie-Woogie in MetaPost, tedious work, and also
scanned the picture. Don't have any use for it as yet.

My case rests, have fun and all the best.

Notes
1. For a survey of biographies of Dutchmathematicians http://bwnw.cwi-incubator.nl/cgi-bin
/uncgi/alf.
2. Born as Piet Mondriaan. At the end of his live, in the USA, he used to call himself Mondrian.
3. Which obeys statistical tests for randomness.
4. A Dutch art movement which began in 1917. Its characteristics are the use of straight lines
and primary colors. The artists were generally after utmost simplicity and abstraction. Main
representatives were Theo van Doesburg, Piet Mondriaan, Bart van der Leck, Gerrit Rietveld,
and J.J.P. Oud, active in various art forms, among others architecture.
5. Adding the year exceeds MetaPost number capacity for the moment.
6. This extra work is superfluous in the PostScript variant.

